apply plugin: 'com.android.application'

android {
    compileSdkVersion 25
    buildToolsVersion "25.0.2"

    defaultConfig {
        applicationId "tv.piratemedia.lightcontroler"
        minSdkVersion 14
        targetSdkVersion 25
        versionCode 1714
        multiDexEnabled true
        versionName "1.8.0 Beta 3"
    }

    if ( project.rootProject.file('signing.properties').isFile() ) {
        Properties signing = new Properties()
        signing.load(project.rootProject.file('signing.properties').newDataInputStream())

        signingConfigs {
            release {
                storeFile project.rootProject.file("signing.jks")
                storePassword signing.getProperty("release.storePass")
                keyAlias signing.getProperty("release.keyUser")
                keyPassword signing.getProperty("release.keyPass")
            }
        }
    }

    buildTypes {
        release {
            minifyEnabled false
            proguardFiles getDefaultProguardFile('proguard-android.txt'), 'proguard-rules.txt'
            if ( project.rootProject.file('signing.properties').isFile() ) {
                signingConfig signingConfigs.release
            }
        }
    }
}

repositories {
    mavenCentral()
    jcenter()
    maven { url "https://jitpack.io" }
}

def playServices = true

dependencies {
    compile fileTree(include: ['*.jar'], dir: 'libs')
    compile 'com.android.support:appcompat-v7:25.+'
    compile 'com.astuetz:pagerslidingtabstrip:1.0.+'
    compile 'com.getpebble:pebblekit:3.0.0'
    compile('com.github.afollestad.material-dialogs:core:0.8.5.3@aar') {
        transitive = true
    }
    if (playServices) {
        compile 'com.google.android.gms:play-services:+'
        wearApp project(':wear');
    }
    compile project(':circularseekbar')
    compile 'com.heinrichreimersoftware.materialdrawer:library:1.4.1'
}
